

```

%_mprintto;
options notes nosource;
proc datasets lib=work nolist memtype=data kill; quit;
%put NOTE:
=====;
%put NOTE: Covance Study Number : 000000106324;
%put NOTE: Client Protocol ID   : ZRHR-REXC-03-EU;
%put NOTE: Program Name        : t_pkconc.sas;
%put NOTE: Purpose              : table of plasma nicotine
concentrations;
%put NOTE: ;
%put NOTE: Input Data           : ADAM.ADPC ADAM.ADSL;
%put NOTE: Output               : t_15_2_4_35(conc);
%put NOTE: Macros Called        : _MPRINTTO;
%put NOTE: ;
%put NOTE: Programmed by        : cvn_jhardman;
%put NOTE: Creation Date        : 2014-07-23;
%put NOTE: SAS Version          : 9.3;
%put NOTE: ;
%put NOTE: == Latest Run
=====;
%put NOTE: Run by                : &sysuserid;
%put NOTE: Date/Time             :
%sysfunc(putn(%sysfunc(date()),e8601da.))T%sysfunc(putn(%sysfunc(time()),
e86011z.));
%put NOTE: ;
%put NOTE: == Modification History
=====;
%put NOTE: Date      Initials   No. Reason;
%put NOTE: 05Aug2014   JMH       1)  Removed Geometric from CI label;
%put NOTE: 16Sep2014   JR        2)  Amended code to reference ANL02FL
and calculate correct change;
%put NOTE: 16Sep2014   JR        3)  Removed geometric for chg from
baseline;
%put NOTE: 17Sep2014   JR        4)  Amended order of mean;
%put NOTE: 19Sep2014   KB        5)  Amended baseline footnote;
%put NOTE: 19Sep2014   KB        6)  Amended ordering for dual prog
table;
%put NOTE: 25Sep2014   JR        7)  Amedned calc of percent for BLOQ;
%put NOTE: 17Oct2014   KB        8)  Added scheduled time for SA Day 5 &
6;
%put NOTE: ;
%put NOTE:
=====;
options notes source source2 nofullstimer validvarname=upcase missing='
';
ods _all_ close;
ods listing;

*=====;
* START OF PROGRAM CODE ;
*=====;

%let tflno=T_15_02_04_35(conc);

```

```

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/));

data _null_;
    tmp="&TFL_Part";
    if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod");
    call symput('TFLpath', compress("&_SASPROGRAMFILE",""));
run;

*****;
* read in data ;
*****;

/*Use ADSL to get N values for column headers*/
data adsl;
    set adam.adsl(where=(fasfl='Y'));
run;

proc sort data=adsl nodupkey out=adsl1;
    by trt01an trt01a subjid;
run;

proc freq data=adsl1(where=(not missing(trt01an))) noprint;
    table trt01an*trt01a/ out =tot(drop=percent rename=(count=total));
run;

data tot2;
    set tot;
    call symput('trt' || compress(put(trt01an,best.)),
compress(total));
run;

/*Bring in appropriate data from adpc*/
data adpc;
    set adam.adpc(where=(paramcd='NIC' and anl01fl='Y' and fasfl='Y'
and pcstat ne 'NOT DONE')); /* 2) JR 16Sep2014 */
    if ablfl='Y' then do; avisit='Baseline'; avisitn=100; end;
    if avisit ne 'Baseline' and avisitn lt 101 then delete;
run;

data adpc_orig;
    set adpc;
    statval=aval;
    type='abs';
    output;
    statval=pchg;
    type='pchg';
    output;
run;

proc sort data=adpc_orig;
    by type trtan trta avisitn avisit atptn atpt;
run;

```

```

proc means data=adpc_orig noprint;
    var statval;
    by type trtan trta avisitn avisit atptn atpt;
    output out=results02 n=n1 mean=mean1 std=std1 median=median1 min=min1
max=max1 q1=q11 q3=q31 lclm=lci1 uclm=uci1;
run;

/* Start 2) JR 16Sep2014 */
DATA ADPC1;
    SET ADAM.ADPC(WHERE=(PARAMCD='NIC' AND ANL02FL='Y' AND FASFL='Y'
AND PCSTAT NE 'NOT DONE') drop=ATPT ATPTN);
    ATTRIB ATPT LENGTH=$40. ATPTN FORMAT=8.;
    ATPT = 'Day 5, sample closest to 08:00 PM';
    ATPTN = 11;
RUN;

DATA ADPC2;
    SET ADPC1;
    STATVAL=PCHG;
    TYPE='pch';
RUN;

PROC SORT DATA=ADPC2;
    BY TYPE TRTAN TRTA AVISITN AVISIT ATPTN ATPT;
RUN;

PROC MEANS DATA=ADPC2 NOPRINT;
    VAR STATVAL;
    BY TYPE TRTAN TRTA AVISITN AVISIT ATPTN ATPT;
    OUTPUT OUT=RESULTS02A N=N1 MEAN=MEAN1 STD=STD1 MEDIAN=MEDIAN1
MIN=MIN1 MAX=MAX1 Q1=Q11 Q3=Q31 LCLM=LCI1 UCLM=UCI1;
RUN;

data results03;
    set results02 RESULTS02A; /* end 2) JR 16Sep2014 */
    attrib meansd length=$20.
        minmax length=$20.
        n length=$20.
        median length=$20.
        quart length=$20.;

    n = left(compress(put(n1,8.)));

    if not missing(median1) then median =
left(compress(put(round(median1,0.01),8.2)));
    if not missing(mean1) and not missing(std1) then meansd =
left(compress(put(round(mean1,0.01),8.2))) || ' (' ||
left(compress(put(0.001*ceil(std1/0.001),8.3))) || ')';
    if not missing(min1) and not missing(max1) then minmax =
left(compress(put(round(min1,0.1),8.1))) || ', ' ||
left(compress(put(round(max1,0.1),8.1)));
    if not missing(lci1) and not missing(uci1) then aci =
strip(strip(put(0.01*floor(lci1/0.01),8.2)) || ', ' ||
strip(put(0.01*ceil(uci1/0.01),8.2)));

```

```

    if not missing(q11) and not missing(q31) then quart =
strip(strip(put(0.01*floor(q11/0.01),8.2)) || ', ' ||
strip(put(0.01*ceil(q31/0.01),8.2)));

    drop /*n1*/ mean1 std1 median1 min1 max1 q11 q31;    /* 7) JR 25Sep2014
*/
run;
%macro outrtf(blankn=, halfblnk=);

%if &halfblnk=N %then %let halfblnk=;
%else %if &halfblnk=Y %then %let halfblnk=~;

/*Obtain subjects with values BLOQ*/
data adpc_blq;
    set adpc;
    where bloqfl='Y';
    statval=aval;
    type='abs';
    output;
    statsval=pchg;
    type='pch';
    output;
run;

proc freq data=adpc_blq noprint;
    table type*trtan*trta*avisitn*avisit*atptn*atpt/ out
=blq(drop=percent /*rename=(trta=trt01a trtan=trt01an)*/);    /* 7) JR
25Sep2014 */
run;

proc sort data=blq;
    by /*trt01an trt01a*/TYPE TRTAN TRTA AVISITN AVISIT ATPTN ATPT; /*
7) JR 25Sep2014 */
run;

%let dsid=%sysfunc(open(blq));
%let nsum=%sysfunc(attrn(&dsid.,nobs));
%let rc=%sysfunc(close(&dsid.));

%put "Check " &nsum.;

%if &nsum. lt 1 %then %do;
    proc sort data=adpc_orig nodupkey out=tpts(keep=type avisitn
avisit atptn atpt trtan trta);
        by type trtan trta avisitn avisit atptn atpt;
    run;

    data blq1;
        set tpts;
        attrib blq length=$50.;
        blq='0';
    run;

%end;

```

```

%else %do;
  /* start 7) JR 25Sep2014 */
  PROC SORT DATA=ADPC_ORIG NODUPKEY OUT=TPTS(KEEP=TYPE AVISITN AVISIT
ATPTN ATPT TRTAN TRTA);
    BY TRTAN TRTA TYPE AVISITN AVISIT ATPTN ATPT;
  RUN;
  DATA BLQTOTS;
    SET RESULTS03(RENAME=(N1=TOTAL));
    KEEP TYPE TR: AVISIT: ATPT: TOTAL;
  RUN;

  PROC SORT DATA=BLQTOTS; BY TYPE TRTAN TRTA AVISITN AVISIT
ATPTN ATPT; RUN;
  PROC SORT DATA=TPTS; BY TYPE TRTAN TRTA AVISITN AVISIT ATPTN
ATPT; RUN;

  DATA TOT_BLQ;
    MERGE TPTS BLQTOTS;
    BY TYPE TRTAN TRTA AVISITN AVISIT ATPTN ATPT;
  RUN;
  PROC SORT DATA=TOT_BLQ;
    BY TYPE TRTAN TRTA AVISITN AVISIT ATPTN ATPT;
  RUN;
  /* end 7) JR 25Sep2014 */

  data blq1;
    attrib blq length=$50.;
    merge blq(in=a) tot_BLQ; /* 7) JR 25Sep2014 */
  /*
  by trt01an trt01a;*/
  BY TYPE TRTAN TRTA AVISITN AVISIT ATPTN ATPT; /* 7) JR
25Sep2014 */

  if not a then do;
    count=0;
  end;
  IF TOTAL NE 0 THEN percent=count/total*100; /* 7) JR
25Sep2014 */

  ELSE PERCENT=0; /* 7) JR 25Sep2014 */

  if count=0 then blq='0';
  else if percent=100 then blq= put(count,3.)||' (100%)';
  else blq=put(count,3.)||'
('||left(strip(put(round(percent,0.1),5.1)))||'%)';
  /*
  rename trt01an=trtan trt01a=trta; */
  run;

  proc sort data=blq1;
    by trtan trta type avisitn avisit atptn atpt;
  run;

%end;

data gmean;
  set adpc_orig;
  if type ne 'pch' then do;
    statvall=statval;

```

```

        if statval ne 0 then ln_statvall=log(statvall);
    end;
run;

proc sort data=gmean;
    by trtan trta type avisitn avisit atptn atpt;
run;

proc means data=gmean noprint;
    output out=gmean1 mean=mean std=std1 lclm=lci1 uclm=uci1 nmiss=miss;
    var ln_statvall;
    by trtan trta type avisitn avisit atptn atpt;
run;

data gmean2;
    set gmean1;
    gmean1=exp(mean);
    gmean=left(compress(put(round(gmean1,0.01),10.2)));
    gcv=compress(put(0.01*ceil((sqrt(exp(std1*std1)-1)*100)/0.01),10.2));
    glci=exp(lci1);
    guci=exp(uci1);
    keep TYPE trtan trta avisitn avisit atptn atpt gmean mean gcv glci guci
    std1 miss;
run;

proc sort data=results03;
    by trtan trta TYPE avisitn avisit atptn atpt;
run;

data results04;
    merge results03 gmean2 blq1;
    attrib gmeancv length=$30.;
    by trtan trta type avisitn avisit atptn atpt;
    if not missing(gcv) then gmeancv=left(trim(gmean)) || ' (' ||
left(trim(gcv))||'%)';
    else gmeancv=left(trim(gmean));
    if not missing(glci) and not missing(guci) then ci =
strip(strip(put(0.01*floor(glci/0.01),10.2)) || ', ' ||
strip(put(0.01*ceil(guci/0.01),10.2)));
run;

proc sort data=results04;
    by type avisitn avisit atptn atpt;
run;

proc transpose data=results04 out=results05 prefix=r name=varname;
    by type avisitn avisit atptn atpt;
    var n meansd median minmax quart blq gmeancv ci aci;
    id trtan;
    idlabel trta;
run;

proc sort data=results05;

```

```

    by type avisitn avisit atptn atpt varname;
run;

data results06;
    set results05;
    attrib stat variable length = $200.;
    varname=upcase(varname);

    if not missing(atpt) then variable=compbl(strip(avisit) ||', '||
atpt);
    else variable=compbl(avisit);

    if varname='N' then do;
        statord=1;
        stat='n';
    end;
    else if varname='BLQ' then do;
        statord=2;
        stat='BLOQ - n (%)';
    end;
    else if varname='GMEANCV' then do;
        statord=3;
        stat='Geometric Mean (CV%)';
    end;
    else if varname='CI' then do;
        statord=4;
        /*      stat='Geometric 95% CI'; */
        STAT='95% CI'; /* 1) JMH 05Aug2014 */
        IF TYPE='pch' THEN DELETE; /* 3) JR 16Sep2014 */
    end;
    else if varname='MEDIAN' then do;
        statord=5;
        stat='Median';
    end;
    else if varname='QUART' then do;
        statord=6;
        stat='Q25, Q75';
    end;
    else if varname='MINMAX' then do;
        statord=7;
        stat='Min, Max';
    end;
    else if varname='MEANS' then do;
        IF TYPE='pch' THEN STATORD=3.5; /* 4) JR 17Sep2014 */
        ELSE statord=8;
        stat='Mean (SD)';
    end;
    else if varname='ACI' then do;
        statord=9;
        stat='95% CI';
        if type ne 'pch' then delete;
    end;
    else put "WA" "RNING: Unexpected varname, please assign stat
and statord " varname= ;

```

```

        drop varname;
run;

data results07;
    set results06;

    if stat='n' or index(stat,'BLOQ') then do;
        if missing(r1) then r1='0';
        if missing(r2) then r2='0';
        if missing(r3) then r3='0';
    end;

    /*ATPTN is not populated for SA arm so we want blanks not
    zeros for BLQ counts*/
    if atptn in (0 1 2 3 4 5 6 7 8 9 10) and index(stat,'BLOQ') then
    r3='';

    /*Similarly for Day 5 and Day 6, if ATPTN is missing, it
    means CC and THS had no results so they should be blank*/
    if (avisit='Day 5' or avisit='Day 6/Discharge') and atptn=. then do;
        if stat='BLOQ - n (%)' then do;
            r1='';
            r2='';
        end;
    end;

    /*We only want the change from baseline for Day 5, 16 hr as
    per the SAP*/
    if type='pch' and atptn ne /*8*/11 then delete; /* 2) JR 16Sep2014 */

    if type='pch' then do;
        atptn=/*8*/11; /* 2) JR 16Sep2014 */
        variable="% Change from Baseline at Day 5, sample closest to
08:00 PM"/*"% Change from Baseline at Day 5 T${sub 0} + 16 h"*/; /* 2) JR
16Sep2014 */

        if stat in ('BLOQ - n (%)' 'Geometric 95% CI' 'Geometric Mean
(CV%)') then delete;
    end;

run;

data labels;
set results07;
    attrib r1 label = "THS 2.2$(N=&trt1)"
           r2 label = "CC$(N=&trt2)"
           r3 label = "SA$(N=&trt3)"
           variable label= "Formatted timepoint"
           variabl1 label= "Unformatted timepoint";

    variabl1=variable;

    flag=1;

```



```

        if index(variable,'T0') then variable=tranwrd(variable,'T0',"T${sub
0}");

        IF MISSING(ATPT) AND AVISITN=105 THEN VARIABLE = 'Day 5 , 08:00-
10:00 PM'; /* 8) KB 17Oct2014 */
        IF MISSING(ATPT) AND AVISITN=106 THEN VARIABLE = 'Day 6/Discharge ,
08:00-10:00 AM'; /* 8) KB 17Oct2014 */

run;

proc sql noprint;
    create table table.T_15_02_04_35 as
    select variable, variabl1, stat, r1, r2, r3
    from labels
    order by /*atptn*/AVISITN, ATPTN, statord; /* 6) KB 19Sep2014 */
quit;

proc sort data=labels;
    by avisitn atptn statord;
run;

data paging;
    set labels;
    by avisitn atptn statord;
    if first.atptn then ln=1; /*Amend to look presentable, and avoid
page overflows*/
    else ln+1;
    if ln=1 then page+1;
    call symput("page",compress(put(page,best.)));
run;

options number nodate orientation=landscape papersize=&p_pgsize missing='
';
ods escapechar='$';
%let linetop = \brdrt\brdrs\brdrw30; * needs to be 1.5pt so calculated
in twips (1/20 pt) ;
%let linebot = \brdrb\brdrs\brdrw30;

ods path stdlib.t106324 (read) ;
ods results off;
ods rtf toc_data
file="/cvn/projects/prj/data/000000106324/TFL/&TFL_Part./&tflno..rtf"
style=t106324 startpage=yes headery=1440 footery=1440 ;
ods noproctitle;
%do i=1 %to &page;

title ;
footnote;
%let nc=0;
%LET GEO=0;

```

```

data comp;
    set paging end=eof;
    where page=&i;

    /* Amend title as needed */
    _firtitl="Table 15.2.4.35 Descriptive Statistics of Plasma Nicotine
Concentrations (ng/mL) - FAS";
    _upcas=(length("Path: &TFLpath.")-
length(compress("Path:&TFLpath.",'ABCDEFGHIJKLMNOPQRSTUVWXYZ')))/2;
    len=&blankn.-length("(page &i of &page)");
    if eof then do;
        call symput('_FSRTITL', trim(left(_firtitl)));
        call symput('_blankn', compress(put(len,best.)));
    end;
    drop _firtitl _upcas len;

    if index(r1,'NC') or index(r2,'NC') or index(r3,'NC') then
call symput('NC',1);
    IF TYPE='abs' then call symput('GEO',1); /* 3) JR 16Sep2014
*/
run;

ods listing close;
ods proclabel = ' ';
* most set up in template others below;
* title arial 12pt bold with 12pt paragraph space below;
* all headers to be arial 11pt bold;
* data arial 10pt;
* headers to be central, text values left aligned and numeric centered
around decimal point;
proc report data = comp missing headline headskip missing nowd split =
'$' %if &i=1 %then %do; contents=' ' %end; %else %do; contents='' %end;;
    column flag page atptn variable statord stat r1 r2 r3;

    define flag          / order order = internal noprint;
    define page          / order order = internal noprint;
    define atptn         / order order=internal noprint;
    define variable      / group style={just=left cellwidth=1.7cm}
style(header)={just=center} "Timepoint";
    define statord       / order order = internal noprint;
    define stat          / display style={just=left cellwidth=2cm}
style(header)={just=center} "Statistic";
    define r1            / display style={just=c cellwidth=1.5cm}
style(header)={just=center};
    define r2            / display style={just=c cellwidth=1.5cm}
style(header)={just=center};
    define r3            / display style={just=c cellwidth=1.5cm}
style(header)={just=center};

    break before flag / page %if &i=1 %then %do;
    contents("&_fsrtitl" %end; %else %do; contents='' %end;;

```

```

break after page / page;

compute after atptn;
  line " ";
endcomp;

compute before page / style={protectspecialchars=off};
  line "&linetop";
endcomp;

compute before _page_ / style={just=left protectspecialchars=off};
  line "\b\fs24\sa24&_FSRTITL." ; * \b = bold, \fs24 is font
size 12pt, \sa24 is space after 12pt;
  line "&linebot";
endcomp;

compute after _page_/ style={just=left protectspecialchars=off
pretext="&LINETOP."};

  line 'Note: CC = Conventional cigarettes; SA = Smoking abstinence;
THS = Tobacco Heating System.';
  %IF &GEO=1 %THEN %DO; /* 3) JR 16Sep2014 */
  line 'Note: Geometric: mean, CV% and 95% confidence interval (CI)
are reported.';
  %END;
  %if &nc=1 %then %do;
    line "Note: NC = Not calculated";
  %end;
  LINE 'Note: T${sub 0} = Time of the first product use on Day
5.';
  LINE 'Note: BLOQ values are imputed by LLOQ/2.';
/*  LINE "Note: Baseline is Day 0, 08:00 PM - 10:00 PM."; */
  LINE 'Note: Baseline is the last assessment prior to first
product use in CC/THS 2.2 arms on Day 1 or last assessment prior to 06:29
AM in SA arm on Day 1.'; /* 5) KB 19Sep2014 */
  line ' ';
  line 'Appendix 15.3.3.4';
  line "Path: &TFLpath." &_blankn.*"\~\~" "(Page &i of &page)";
  line "Program Run: &sysdate &sysuserid Program Status:
&status";
endcomp;

run;
%end;
ods rtf close;
ods results on;
ods path sashelp.tmplmst (read);

%mend ;

%outrtf(blankn=70, halfblnk=N);
ods listing;
proc printto print = "&table./T_15_02_04_35.lst" new;
run;

```

```
proc contents data = table.T_15_02_04_35 varnum;  
run;  
ods listing close;  
proc printto ; run;  
*=====;  
* END OF PROGRAM CODE ;  
*=====;
```