
%_mprintto; 

options notes nosource; 

proc datasets lib=work nolist memtype=data kill; quit; 

%put NOTE: 

=======================================================================; 

%put NOTE: Covance Study Number : 000000106324; 

%put NOTE: Client Protocol ID   : ZRHR-REXC-03-EU; 

%put NOTE: Program Name         : t_pkparam.sas; 

%put NOTE: Purpose              : table of  nicotine concentrations; 

%put NOTE: ; 

%put NOTE: Input Data           : ADAM.ADPP ADAM.ADSL; 

%put NOTE: Output               : t_15_2_4_38(pkparam2); 

%put NOTE: Macros Called        : _MPRINTTO; 

%put NOTE: ; 

%put NOTE: Programmed by        : cvn_jhardman; 

%put NOTE: Creation Date        : 2014-08-04; 

%put NOTE: SAS Version          : 9.3; 

%put NOTE: ; 

%put NOTE: == Latest Run 

=========================================================; 

%put NOTE: Run by               : &sysuserid; 

%put NOTE: Date/Time            : 

%sysfunc(putn(%sysfunc(date()),e8601da.))T%sysfunc(putn(%sysfunc(time()),

e8601lz.)); 

%put NOTE: ; 

%put NOTE: == Modification History 

===============================================; 

%put NOTE: Date        Initials   No. Reason; 

%put NOTE: ; 

%put NOTE: 

=======================================================================; 

options notes source source2 nofullstimer validvarname=upcase missing=' 

'; 

ods _all_ close; 

ods listing; 

 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

 

%let tflno=T_15_02_04_38(pkparam2); 

 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 

  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

run; 

 

******************************; 

* read in data ; 

******************************; 

 



/*Use ADSL to get N values for column headers*/ 

data adsl; 

 set adam.adsl(where=(fasfl='Y')); 

run; 

 

proc sort data=adsl nodupkey out=adsl1; 

 by trt01an trt01a subjid; 

run; 

 

proc freq data=adsl1(where=(not missing(trt01an))) noprint;   

 table trt01an*trt01a/ out =tot(drop=percent rename=(count=total)); 

run; 

 

data tot2; 

 set tot; 

 call symput('trt' || compress(put(trt01an,best.)), 

compress(total)); 

run; 

 

/*Bring in appropriate data from adpp*/ 

data adpp; 

 set adam.adpp(where=(parcat1='Cotinine' and anl01fl='Y' and 

fasfl='Y'));  

 

    param=tranwrd(param,'(Cotinine)',''); 

run; 

 

data adpp_orig;  

 set adpp; 

 statval=aval; 

run; 

 

proc sort data=adpp_orig; 

    by trtan trta paramn param; 

run; 

 

 

proc means data=adpp_orig noprint; 

    var statval; 

    by trtan trta paramn param; 

    output out=results02 n=n1 mean=mean1 std=std1 median=median1 min=min1 

max=max1 q1=q11 q3=q31 lclm=lci1 uclm=uci1; 

run; 

 

data results03; 

    set results02; 

    attrib meansd length=$20. 

         minmax length=$20. 

         n      length=$20. 

         median length=$20. 

          quart  length=$20.; 

 

    n = left(compress(put(n1,8.))); 

 



 if not missing(median1) then median  = 

left(compress(put(round(median1,0.01),8.2))); 

 if not missing(mean1) and not missing(std1) then meansd  = 

left(compress(put(round(mean1,0.01),8.2))) || ' (' || 

left(compress(put(0.001*ceil(std1/0.001),8.3))) || ')'; 

 if not missing(min1) and not missing(max1) then minmax  = 

left(compress(put(round(min1,0.1),8.1))) || ', ' || 

left(compress(put(round(max1,0.1),8.1))); 

 if not missing(lci1) and not missing(uci1) then ci = 

strip(strip(put(0.01*floor(lci1/0.01),8.2)) || ', ' || 

strip(put(0.01*ceil(uci1/0.01),8.2)));  

    if not missing(q11) and not missing(q31) then quart = 

strip(strip(put(0.01*floor(q11/0.01),8.2)) || ', ' || 

strip(put(0.01*ceil(q31/0.01),8.2))); 

 

    drop n1 mean1 std1 median1 min1 max1 q11 q31;   

run; 

 

data gmean; 

 set adpp_orig; 

 statval1=statval; 

 if statval ne 0 then ln_statval1=log(statval1); 

run; 

 

proc sort data=gmean; 

    by trtan trta paramn param; 

run; 

 

proc means data=gmean noprint; 

  output out=gmean1 mean=mean std=std1 lclm=lci1 uclm=uci1 nmiss=miss; 

  var ln_statval1; 

    by trtan trta paramn param; 

run; 

 

data gmean2; 

  set gmean1; 

   gmean1=exp(mean); 

   gmean=left(compress(put(round(gmean1,0.01),10.2))); 

   gcv=compress(put(0.01*ceil((sqrt(exp(std1*std1)-1)*100)/0.01),10.2));  

   glci=exp(lci1); 

   guci=exp(uci1); 

  keep trtan trta paramn param gmean mean gcv glci guci std1 miss; 

run; 

 

proc sort data=results03; 

    by trtan trta paramn param; 

run; 

 

data results04;  

  merge results03 gmean2; 

 attrib gmeancv length=$30.; 

    by trtan trta paramn param; 

  if not missing(gcv) then gmeancv=left(trim(gmean)) || ' (' || 

left(trim(gcv))||'%)';  



  else gmeancv=left(trim(gmean)); 

  if not missing(glci) and not missing(guci) then ci = 

strip(strip(put(0.01*floor(glci/0.01),10.2)) || ', ' || 

strip(put(0.01*ceil(guci/0.01),10.2)));  

run; 

 

proc sort data=results04; 

    by paramn param; 

run; 

 

proc transpose data=results04 out=results06 prefix=r name=varname; 

    by paramn param; 

    var n meansd median minmax quart gmeancv ci; 

    id trtan; 

    idlabel trta; 

run; 

 

proc sort data=results06; 

    by paramn param; 

run; 

 

data results07; 

    set results06; 

    attrib stat length = $200.; 

    varname=upcase(varname);  

 

    if varname='N' then do;  

        statord=1;  

        stat='n';  

    end; 

 else if varname='GMEANCV' then do; 

        statord=2;  

        stat='Geometric Mean (CV%)';  

 end; 

    else if varname='CI' then do;  

        statord=3;  

        stat='95% CI';  

    end; 

    else if varname='MEDIAN' then do;  

        statord=4;  

        stat='Median';  

    end; 

    else if varname='QUART' then do; 

        statord=5; 

        stat='Q25, Q75'; 

    end; 

    else if varname='MINMAX' then do;  

        statord=6;  

        stat='Min, Max';  

    end; 

    else if varname='MEANSD' then do;  

        statord=7;  

        stat='Mean (SD)';  

    end; 



    drop varname; 

run; 

 

data results08; 

    set results07; 

 

    if stat='N' or stat='n' then do; 

        if missing(r1) then r1='0'; 

        if missing(r2) then r2='0'; 

    end;     

run; 

 

data labels; 

set results08; 

    attrib r1 label = "THS 2.2$(N=&trt1)" 

           r2 label = "CC$(N=&trt2)" 

           param1 label="Unformatted parameter" 

           param label="Formatted parameter" 

           paramn label="Original parameter number"   

           paramn2 label="Sorting parameter number";  

 

   flag=1; 

 

            param1=param; 

 

            if paramn=4 then param=tranwrd(param,"Average Conc","C`{sub 

avg}"); 

            else if paramn=5 then param=tranwrd(param,"Max Conc","C`{sub 

peak}");  

            else if paramn=6 then param=tranwrd(param,"Time of 

CMAX","t`{sub peak}");  

  

            if paramn=4 then paramn2=6; 

            else if paramn=5 then paramn2=4; 

            else if paramn=6 then paramn2=5; 

 

            /*Consistent with PK-02, no gmean, gcv, mean, sd, gci for 

tmax*/ 

            if paramn=6 then do; 

                if statord in (2,3,7) then delete; 

            end; 

 

            if statord=7 then delete;  

 

 

run; 

 

proc sql noprint; 

 create table table.t_15_02_04_38 as 

 select param, stat, r1, r2 

    from labels 

 order by paramn2, statord;   

quit; 

 



proc sort data=labels; 

  by paramn2 statord;   

run; 

 

data paging; 

    set labels; 

  by paramn2 statord;   

    if  first.paramn2 and ln gt 11 then ln=1; /*Amend to look 

presentable, and avoid page overflows*/  

    else ln+1; 

    if ln=1 then page+1; 

    call symput("page",compress(put(page,best.))); 

run; 

 

%macro outrtf(blankn=, halfblnk=); 

 

%if &halfblnk=N %then %let halfblnk=; 

%else %if &halfblnk=Y %then %let halfblnk=\~; 

 

options number nodate orientation=landscape papersize=&p_pgsize missing=' 

'; 

ods escapechar='`'; 

%let linetop = \brdrt\brdrs\brdrw30;  * needs to be 1.5pt so calculated 

in twips (1/20 pt) ; 

%let linebot = \brdrb\brdrs\brdrw30; 

 

ods path stdlib.t106324 (read) ; 

ods results off; 

ods rtf toc_data 

file="/cvn/projects/prj/data/000000106324/TFL/&TFL_Part./&tflno..rtf" 

style=t106324 startpage=yes headery=1440 footery=1440 ; 

ods noproctitle; 

%do i=1 %to &page; 

 

 

title ; 

footnote; 

%let nc=0; 

 

data comp; 

     set paging end=eof;   

  where page=&i; 

      

    /* Amend title as needed */ 

    _firtitl="Table 15.2.4.38 Descriptive Statistics of Plasma Cotinine 

Concentration Parameters on Day 5 - FAS";   

    _upcas=(length("Path: &TFLpath.")-

length(compress("Path:&TFLpath.",'ABCDEFGHIJKLMNOPQRSTUVWXYZ')))/2; 

    len=&blankn.-length("(page &i of &page)");  

    if eof then do; 

    call symput('_FSRTITL', trim(left(_firtitl))); 

    call symput('_blankn', compress(put(len,best.))); 

    end; 

    drop _firtitl  _upcas len; 



run; 

 

 

ods listing close; 

ods proclabel = ' '; 

* most set up in template others below; 

* title arial 12pt bold with 12pt paragraph space below; 

* all headers to be arial 11pt bold; 

* data arial 10pt; 

* headers to be central, text values left aligned and numeric centered 

around decimal point; 

proc report data = comp missing headline headskip missing nowd split = 

'$' %if &i=1 %then %do; contents=' ' %end; %else %do; contents='' %end;; 

 column flag page paramn2 param statord stat r1 r2;    

 

    define flag         / order order = internal noprint; 

   define page         / order order = internal noprint; 

    define paramn2        / order order=internal noprint; 

    define param         / group style={just=left cellwidth=2cm} 

style(header)={just=center} "Parameter"; 

    define statord      / order order = internal noprint; 

    define stat         / display style={just=left cellwidth=2cm} 

style(header)={just=center} "Statistic"; 

    define r1           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center};   

    define r2           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center};   

 

 break before flag / page %if &i=1 %then %do;  

 contents="&_fsrtitl" %end; %else %do; contents='' %end;;  

 

 

 break after page / page; 

 

 compute after paramn2; 

  line " "; 

 endcomp; 

 

 compute before page / style={protectspecialchars=off};; 

     line "&linetop"; 

  endcomp; 

 

  compute before _page_ / style={just=left protectspecialchars=off};  

     line "\b\fs24\sa24&_FSRTITL." ;  * \b = bold, \fs24 is font 

size 12pt, \sa24 is space after 12pt; 

  line "&linebot"; 

  endcomp; 

 

  compute after _page_/ style={just=left protectspecialchars=off 

pretext="&LINETOP."}; 

 

 line 'Note: CC = Conventional cigarettes; THS = Tobacco Heating 

System.';  



 line 'Note: Geometric: mean, CV% and 95% confidence interval (CI) 

are reported.';  

  line ' '; 

    LINE "Appendix 15.3.3.7";   

  line "Path: &TFLpath." &_blankn.*"\~\~" "(Page &i of &page)"; 

  line "Program Run: &sysdate  &sysuserid   Program Status: 

&status"; 

  endcomp; 

run; 

%end; 

ods rtf close; 

ods results on; 

ods path sashelp.tmplmst (read); 

 

%mend ; 

 

%outrtf(blankn=69, halfblnk=N); 

ods listing; 

proc printto print = "&table./T_15_02_04_38.lst" new; 

run; 

 

proc contents data = table.T_15_02_04_38 varnum; 

run; 

ods listing close; 

proc printto ; run; 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 

 

 

 


