
%_mprintto; 

options notes nosource; 

proc datasets lib=work nolist memtype=data kill; quit; 

%put NOTE: 

=======================================================================; 

%put NOTE: Covance Study Number : 000000106324; 

%put NOTE: Client Protocol ID   : ZRHR-REXC-03-EU; 

%put NOTE: Program Name         : t_qsu.sas; 

%put NOTE: Purpose              : table of QSU factors and total score; 

%put NOTE: ; 

%put NOTE: Input Data           : ADAM.ADSL ADAM.ADQSSU; 

%put NOTE: Output               : t_15_2_4_43(qsu); 

%put NOTE: Macros Called        : _MPRINTTO; 

%put NOTE: ; 

%put NOTE: Programmed by        : cvn_jhardman; 

%put NOTE: Creation Date        : 2014-07-30; 

%put NOTE: SAS Version          : 9.3; 

%put NOTE: ; 

%put NOTE: == Latest Run 

=========================================================; 

%put NOTE: Run by               : &sysuserid; 

%put NOTE: Date/Time            : 

%sysfunc(putn(%sysfunc(date()),e8601da.))T%sysfunc(putn(%sysfunc(time()),

e8601lz.)); 

%put NOTE: ; 

%put NOTE: == Modification History 

===============================================; 

%put NOTE: Date        Initials   No. Reason; 

%put NOTE: 01Aug2014   JMH        1)  Applied baseline update;  

%put NOTE: 16Sep2014   JMH        2) Added footnote as per client 

comments;  

%put NOTE: 18Sep2014   JR         3) Amended baseline footnote;  

%put NOTE: 19Sep2014   KB         4) Amended quartiles to match other 

tables;  

%put NOTE: 22Sep2014   KB         5) Added update 5 to the percentage 

change;  

%put NOTE: 

=======================================================================; 

options notes source source2 nofullstimer validvarname=upcase missing=' 

'; 

ods _all_ close; 

ods listing; 

 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

 

%let tflno=T_15_02_04_43(qsu); 

 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 



  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

run; 

 

******************************; 

* read in data ; 

******************************; 

 

/* Calculate totals for products */ 

data adsl; 

 set adam.adsl(where=(pprotfl='Y' and fasfl='Y')); 

 analgr1n=1; 

 analgr1='FAS'; 

 trtord=trt01an; 

run; 

 

proc sort data=adsl nodupkey out=adsl1; 

 by analgr1n analgr1 trtord subjid; 

run; 

 

proc freq data=adsl1(where=(not missing(trtord))) noprint;   

 table analgr1n*analgr1*trtord/ out =totals2(drop=percent 

rename=(count=total)); 

run; 

 

data totals3; 

    set totals2; 

 

    call symput('trt'||strip(put(trtord,best.)),strip(put(total,best.))); 

run; 

 

proc sort data=totals3; 

    by analgr1n analgr1 trtord; 

run; 

 

/*Bring in appropriate data from  ADQSSU*/ 

data adqssu; 

 set adam.adqssu(where=(anl01fl='Y' and pprotfl='Y' and fasfl='Y'));  

 IF ABLFL='Y' THEN DO; AVISIT='Baseline'; AVISITN=100; END; /* 1) 

JMH 01Aug2014 */  

 IF AVISIT NE 'Baseline' AND AVISITN LT 101 THEN DELETE; /* 1) JMH 

01Aug2014 */  

run; 

 

/*Raw values*/ 

data adqssu_orig;  

 set adqssu; 

 analgr1n=1; 

 analgr1='FAS'; 

 

 statval=aval; 

run; 

 

proc sort data=adqssu_orig; 



    by analgr1n analgr1 trtan trta parcat2n parcat2 paramn param paramcd 

avisitn avisit; 

run; 

 

proc means data=adqssu_orig alpha=0.05 noprint; 

    var statval; 

    by analgr1n analgr1 trtan trta parcat2n parcat2 paramn param paramcd 

avisitn avisit; 

    output out=results02_orig n=n1 mean=mean1 std=std1 median=median1 

min=min1 max=max1 q1=q1 q3=q3 lclm=lci1 uclm=uci1; 

run; 

 

data results03_orig; 

    set results02_orig; 

    attrib meansd length=$20. 

        minmax length=$20. 

        n      length=$20. 

        median length=$20. 

         ci     length=$20. 

         quart  length=$20.; 

 

    n = left(compress(put(n1,8.))); 

    if not missing(median1) then median  = 

left(compress(put(round(median1,0.01),8.2)));  

    if not missing(mean1) and not missing(std1) then meansd  = 

left(compress(put(round(mean1,0.01),8.2))) || ' (' || 

left(compress((put(0.001*ceil(std1/0.001),8.3)))) || ')';    

    if not missing(min1) and not missing(max1) then minmax  = 

left(compress(put(round(min1,0.1),8.1))) || ', ' || 

left(compress(put(round(max1,0.1),8.1))); 

    if not missing(lci1) and not missing(uci1) then ci = 

strip(strip(compress(put(0.01*floor(lci1/0.01),8.2)))) || ', ' || 

strip(compress(put(0.01*ceil(uci1/0.01),8.2)));  

/*    if not missing(q1) and not missing(q3) then quart = 

strip(strip(put(round(q1,0.01),8.2)) || ', ' || 

strip(put(round(q3,0.01),8.2)));*/ 

     IF NOT MISSING(Q1) AND NOT MISSING(Q3) THEN QUART = 

STRIP(PUT(0.01*FLOOR(Q1*100),10.2)) || ', ' || 

STRIP(PUT(0.01*CEIL(Q3*100),10.2));    /* 4) KB 19Sep2014 */  

 

    drop n1 mean1 std1 median1 min1 max1 lci1 uci1 q1 q3; 

run; 

 

data results04_orig; /*Create text as required in output*/ 

    set results03_orig; 

    attrib paramc length = $100.; 

 

    if index(parcat2,'Factor')=0 then paramc=strip(param); 

    else paramc=strip(parcat2); 

 

    drop param paramcd parcat2 parcat2n; 

run; 

 

proc sort data=results04_orig; 



     by paramn paramc avisitn avisit; 

run; 

 

proc transpose data=results04_orig out=results05_orig prefix=o 

name=varname; 

    by paramn paramc avisitn avisit; 

    var n meansd median minmax ci quart; 

    id trtan; 

    idlabel trta; 

run; 

/*End of raw values*/ 

 

/*%Change values*/ 

 

data adqssu_chg;  

 set adqssu; 

 analgr1n=1; 

 analgr1='FAS'; 

 

 statval=pchg;  

run; 

 

proc sort data=adqssu_chg; 

    by analgr1n analgr1 trtan trta parcat2n parcat2 paramn param paramcd 

avisitn avisit; 

run; 

 

proc means data=adqssu_chg alpha=0.05 noprint; 

    var statval; 

    by analgr1n analgr1 trtan trta parcat2n parcat2 paramn param paramcd 

avisitn avisit; 

    output out=results02_chg n=n1 mean=mean1 std=std1 median=median1 

min=min1 max=max1 q1=q1 q3=q3 lclm=lci1 uclm=uci1; 

run; 

 

data results03_chg; 

    set results02_chg; 

    attrib meansd length=$20. 

        minmax length=$20. 

        n      length=$20. 

        median length=$20. 

           ci     length=$20. 

           quart  length=$20.; 

 

    n = left(compress(put(n1,8.))); 

    if not missing(median1) then median  = 

left(compress(put(median1,8.2)));  

    if not missing(mean1) and not missing(std1) then meansd  = 

left(compress(put(mean1,8.2))) || ' (' || 

left(compress((put(0.001*ceil(std1/0.001),8.3)))) || ')';    

    if not missing(min1) and not missing(max1) then minmax  = 

left(compress(put(min1,8.1))) || ', ' || left(compress(put(max1,8.1))); 



    if not missing(lci1) and not missing(uci1) then ci = 

strip(strip(compress(put(0.01*floor(lci1/0.01),8.2)))) || ', ' || 

strip(compress(put(0.01*ceil(uci1/0.01),8.2)));  

/*    if not missing(q1) and not missing(q3) then quart = 

strip(strip(put(q1,8.2)) || ', ' || strip(put(q3,8.2)));*/ 

     IF NOT MISSING(Q1) AND NOT MISSING(Q3) THEN QUART = 

STRIP(PUT(0.01*FLOOR(Q1*100),10.2)) || ', ' || 

STRIP(PUT(0.01*CEIL(Q3*100),10.2));  /* 5) KB 22Sep2014 */  

 

 

    drop n1 mean1 std1 median1 min1 max1 lci1 uci1 q1 q3; 

run; 

 

data results04_chg; /*Create text as required in output*/ 

    set results03_chg; 

    attrib paramc length = $100.; 

 

    if index(parcat2,'Factor')=0 then paramc=strip(param); 

    else paramc=strip(parcat2); 

 

    drop param paramcd parcat2 parcat2n; 

run; 

 

proc sort data=results04_chg; 

     by paramn paramc avisitn avisit; 

run; 

 

proc transpose data=results04_chg out=results05_chg prefix=c 

name=varname; 

    by paramn paramc avisitn avisit; 

    var n meansd median minmax ci quart; 

    id trtan; 

    idlabel trta; 

run; 

/*End of %Change values*/ 

 

/*Combine raw and change values*/ 

proc sort data=results05_orig; by paramn paramc avisitn avisit varname; 

run; 

proc sort data=results05_chg; by paramn paramc avisitn avisit varname; 

run; 

 

data results05; 

 merge results05_orig results05_chg; 

  by paramn paramc avisitn avisit varname; 

run; 

 

data results06; 

    set results05; 

    attrib stat length = $100.; 

 

    if varname='N' then do;  

        statord=1;  

        stat='n';  



    end; 

    else if varname='MEANSD' then do;  

        statord=2;  

        stat='Mean (SD)';  

    end; 

    else if varname='CI' then do;  

        statord=3;  

        stat='95% CI';  

    end; 

    else if varname='MEDIAN' then do;  

        statord=4;  

        stat='Median';  

    end; 

    else if varname='QUART' then do; 

        statord=5; 

        stat='Q25, Q75'; 

    end; 

    else if varname='MINMAX' then do;  

        statord=6;  

        stat='Min, Max';  

    end; 

 

    drop varname; 

run; 

 

data results07; 

    set results06; 

 

    if stat='n' then do; 

        if missing(o1) then o1='0'; 

        if missing(o2) then o2='0'; 

        if missing(o3) then o3='0'; 

    end; 

  if avisitn in(99 100) then do; 

    c1=''; 

    c2=''; 

    c3=''; 

  end; 

run; 

 

data labels; 

set results07; 

    attrib o1 label = "Raw value" 

           o2 label = "Raw value" 

           o3 label = "Raw value" 

           c1 label = "% Change(*)" 

           c2 label = "% Change(*)" 

           c3 label = "% Change(*)" 

       avisit label= "Unformatted timepoint"; 

  

   if paramc = 'Total Score' then paramc = 'Total score';  

 

run; 

 



proc sort data=labels; 

  by paramn avisitn statord; 

run; 

 

proc sql noprint; 

 

create table table.t_15_02_04_43 as 

select paramc, avisit, stat, o1, c1, o2, c2, o3, c3 

from labels 

order by paramn, avisitn, statord; 

 

quit; 

 

 

data paging; 

    set labels; 

    by paramn avisitn statord; 

 

    flag=1;  

 

    if ln gt 11 then ln=1; /*Amend to look presentable, and avoid page 

overflows*/ 

    else ln+1; 

    if ln=1 then page+1; 

    call symput("page",compress(put(page,best.))); 

run; 

 

options number nodate orientation=landscape papersize=&p_pgsize missing=' 

'; 

ods escapechar='$'; 

%let linetop = \brdrt\brdrs\brdrw30;  * needs to be 1.5pt so calculated 

in twips (1/20 pt) ; 

%let linebot = \brdrb\brdrs\brdrw30; 

%macro outrtf(blankn=, halfblnk=); 

 

%if &halfblnk=N %then %let halfblnk=; 

%else %if &halfblnk=Y %then %let halfblnk=\~; 

 

 

ods path stdlib.t106324 (read) ; 

ods results off; 

ods rtf toc_data 

file="/cvn/projects/prj/data/000000106324/TFL/&TFL_Part./&tflno..rtf" 

style=t106324 startpage=yes headery=1440 footery=1440 ; 

ods noproctitle; 

%do i=1 %to &page; 

 

 

title ; 

footnote; 

%let wd=0; 

ods proclabel=' '; 

 

 



data comp; 

     set paging end=eof;   

  where page=&i; 

      

    /* Amend title as needed */ 

    _firtitl="Table 15.2.4.43 Descriptive Statistics of QSU-brief 

Questionnaire Factors and Total Score - FAS";   

    _upcas=(length("Path: &TFLpath.")-

length(compress("Path:&TFLpath.",'ABCDEFGHIJKLMNOPQRSTUVWXYZ')))/2; 

    len=&blankn.-length("(page &i of &page)");  

    if eof then do; 

    call symput('_FSRTITL', trim(left(_firtitl))); 

    call symput('_blankn', compress(put(len,best.))); 

    end; 

    drop _firtitl  _upcas len; 

run; 

 

 

ods listing close; 

 

* most set up in template others below; 

* title arial 12pt bold with 12pt paragraph space below; 

* all headers to be arial 11pt bold; 

* data arial 10pt; 

* headers to be central, text values left aligned and numeric centered 

around decimal point; 

proc report data = comp missing headline headskip missing nowd split = 

'$' %if &i=1 %then %do; contents=' ' %end; %else %do; contents='' %end;; 

; 

 column flag page paramn paramc avisitn avisit statord stat ("THS 

2.2$(N=&trt1)&linebot" o1 c1) ("CC$(N=&trt2)&linebot" o2 c2) 

("SA$(N=&trt3)&linebot" o3 c3);  

         

    define flag         / order order=internal noprint;  

    define page         / order order = internal noprint; 

    define paramn       / order order = internal noprint; 

    define paramc       / group style={just=l cellwidth=1.8cm} 

style(header)={just=center} "Variable"; 

    define avisitn      / order order=internal noprint; 

    define avisit       / group style={just=l cellwidth=1.6cm} 

style(header)={just=center} "Timepoint"; 

    define statord      / order order = internal noprint; 

    define stat         / display style={just=l cellwidth=1.5cm} 

style(header)={just=center} "Statistic"; 

    define o1           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center}; 

    define c1           / display style={just=c cellwidth=1.7cm} 

style(header)={just=center}; 

    define o2           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center}; 

    define c2           / display style={just=c cellwidth=1.7cm} 

style(header)={just=center}; 

    define o3           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center}; 



    define c3           / display style={just=c cellwidth=1.7cm} 

style(header)={just=center}; 

 

    break before flag / page %if &i=1 %then %do;  

    contents="&_fsrtitl" %end; %else %do; contents='' %end;;  

 

 break after page / page; 

 

 compute after avisitn; 

  line " "; 

 endcomp; 

 

 compute before page / style={protectspecialchars=off};; 

     line "&linetop"; 

  endcomp; 

 

  compute before _page_ / style={just=left protectspecialchars=off};  

     line "\b\fs24\sa24&_FSRTITL." ;  * \b = bold, \fs24 is font 

size 12pt, \sa24 is space after 12pt; 

  line "&linebot"; 

  endcomp; 

 

  compute after _page_/ style={just=left protectspecialchars=off 

pretext="&linetop."}; 

  line 'Note: CC = Conventional cigarettes; SA = Smoking 

abstinence; THS = Tobacco Heating System.'; 

  line "Note: * % change from baseline, where baseline is the 

last assessment prior to first product use in CC/THS 2.2 arms on Day 1 or 

last assessment prior to 06:29 AM in SA arm on Day 1."; /* 3) JR 

18Sep2014 */  

/*  line "Note: * % change from baseline, where baseline is 

defined as the last assessment prior to 06:29 AM on Day 1.";*/ 

  LINE 'Note: QSU-brief scores reported on a 7-point scale. 

Higher scores indicate greater intensity of urge.'; /* 2) JMH 16Sep2014 

*/  

  line ' '; 

  line 'Appendix 15.3.6.11'; 

  line "Path: &TFLpath." &_blankn.*"\~\~" "(Page &i of &page)"; 

  line "Program Run: &sysdate  &sysuserid   Program Status: 

&status"; 

  endcomp; 

run; 

%end; 

ods rtf close; 

ods results on; 

ods path sashelp.tmplmst (read); 

 

%mend ; 

 

%outrtf(blankn=70, halfblnk=N); 

 

ods listing; 

proc printto print = "&table./t_15_02_04_43.lst" new; 

run; 



 

proc contents data = table.t_15_02_04_43 varnum; 

run; 

ods listing close; 

 

proc printto ; run; 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 

 

 

 


