
%_mprintto; 

options notes nosource; 

proc datasets lib=work nolist memtype=data kill; quit; 

/* Standard - leave this */ 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

/* Standard - leave this */ 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 

  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

run; 

%put NOTE: 

=======================================================================; 

%put NOTE: Covance Study Number : 000000106324; 

%put NOTE: Client Protocol ID   : ZRHR-REXC-03-EU; 

%put NOTE: Program Name         : tl_anlpkparam.sas; 

%put NOTE: Purpose              : table and figure of PK paramters; 

%put NOTE: ; 

%put NOTE: Input Data           : ADAM.ADPP; 

%put NOTE: Output               : L_15_04_04_39(PK) T_15_2_4_39(PK) ; 

%put NOTE: Macros Called        : _MPRINTTO; 

%put NOTE: ; 

%put NOTE: Programmed by        : cvn_ahall; 

%put NOTE: Creation Date        : 2014-30-07; 

%put NOTE: SAS Version          : 9.3; 

%put NOTE: ; 

%put NOTE: == Latest Run 

=========================================================; 

%put NOTE: Run by               : &sysuserid; 

%put NOTE: Date/Time            : 

%sysfunc(putn(%sysfunc(date()),e8601da.))T%sysfunc(putn(%sysfunc(time()),

e8601lz.)); 

%put NOTE: ; 

%put NOTE: == Modification History 

===============================================; 

%put NOTE: Date        Initials   No. Reason; 

%put NOTE: ; 

%put NOTE: 

=======================================================================; 

options notes source source2 nofullstimer validvarname=upcase missing=' ' 

noquotelenmax/*turn off warnings about quoted strings to long*/;; 

ods _all_ close; 

ods listing; 

 

 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

data adsl; 

  set adam.adsl(where=(pprotfl='Y')); 

  if index(trt01a,'THS 2.2') then colord=1; 

  output; 

  if index(trt01a,'CC') then colord=2; 



  output; 

  if index(trt01a,'SA') then colord=3; 

  output; 

run; 

 

proc sort data=adsl nodupkey out=adsl1; 

 by colord subjid; 

run; 

 

proc freq data=adsl1(where=(not missing(colord))) noprint;   

 table colord/ out =totals2(drop=percent rename=(count=total)); 

run; 

 

data _null_; 

    set totals2; 

    call symput('tot'||strip(put(colord,best.)),strip(put(total,best.))); 

run; 

******************************; 

* read in data ; 

 

******************************; 

 

%macro anal(param=, tabno=, title=, reference=); 

 

/*formats macro and appendix output macros*/ 

%include 

"/cvn/projects/prj/development/000000106324/dev/adhoc/TMPLTMIX.sas"; 

 

proc sort data=adam.adpp(where=( 

 not missing(aval) and pprotfl='Y' and anl01fl='Y' and 

upcase(parcat1)="&param")) 

          out=adpp; 

          by subjid paramcd; 

run; 

 

 

data adpp1 noanal; 

  attrib param1 length=$100; 

  set adpp;  

    if  index(paramcd,'CMAX') THEN PARAM1='C`{sub peak} 

('||strip(avalu)||')'; 

    else if  index(paramcd,'TMAX') THEN PARAM1='t`{sub peak} 

('||strip(avalu)||')'; 

    else  if  index(paramcd,'CAVG') THEN PARAM1='C`{sub avg} 

('||strip(avalu)||')'; 

if  index(paramcd,'CMAX') or index(paramcd,'CAVG') then do; 

       if aval not in (.,0) then do; 

            logaval=log(aval); 

            output adpp1; 

    end; 

    else output noanal; 

end; 

else output adpp1; 

run; 



 

/*treatment and parameter formats to display text rather than numbers for 

listing*/ 

%fmt(datain=adpp1, start=paramn, label=param1, name=param); 

%fmt(datain=adpp1, start=trtan, label=trta, name=trt); 

 

 

data adppmod; 

    set adpp1; 

    format paramn param. trtan trt.  ; 

run; 

 

proc sort data=adppmod; by paramn subjid; run; 

 

title1 j=l "PAGESPLIT"; /*do not change*/ 

title2 j=l 'Parameter: #byval1';  

title3 j=l 'Proc Mixed Procedure'; 

TITLE4 J=L "The where clause used on the dataset adam.adpp: pprotfl='Y' 

and anl01fl='Y'"; 

%let tflno=L_15_04_04_&tabno(PK); 

 

 

%mixout1(fileout=/cvn/projects/prj/data/000000106324/TFL/&TFL_Part./&tfln

o); 

options ps=28; 

  

options nobyline; 

 

proc mixed data=adppmod(where=(index(paramcd,'TMAX')=0)) method=reml 

maxiter=200 order=internal; 

    by paramn; 

    class trtan sexc ucpdgr1; 

    model logaval = trtan sexc ucpdgr1 / outp=pred ; 

    lsmeans trtan / pdiff alpha=0.05 cl ; 

    ods output lsmeans=lsmeans; 

    ods output diffs=diffs ; 

 ods output covparms=covparms(rename=(estimate=residual));  

run; 

 

 

/*Residual Plots*/ 

title3 j=l 'Residual Plots'; 

options ps=27; /*change this for proc plot*/ 

 

 proc rank data=pred out=resid normal=vw ; 

 by paramn; 

  ranks nscore; 

  var resid; 

  run; 

 

  proc plot data=resid hpercent=50; 

  by paramn; 

  plot resid*pred / vref=0; 

  plot resid*nscore; 



  run; 

  quit; 

  

/*data counts*/ 

proc univariate data=adppmod(where=(index(paramcd,'TMAX')=0)) noprint; 

   by paramn; 

   class trtan; 

   var logaval; 

   output out=num1 n=n1; 

run; 

 

data diffs1; 

 merge diffs covparms ; 

 by paramn; 

run; 

 

 

data tabout; 

length out $100 stat $100; 

   set lsmeans(in=a) diffs1(in=b) num1(in=c); 

   /*ordering columns of treatments*/ 

   if b then colord=3; 

   else if trtan=1 then colord=1; 

   else if trtan=2 then colord=2; 

 

   /* N row*/ 

    if c then do; 

        ord=1; 

        stat='n'; 

        out=compress(put(n1,best.)); 

        output; 

    end; 

 

    if a then do; 

    /*Back transformation*/ 

    estimatee=exp(estimate); 

    lowere=exp(lower); 

    uppere=exp(upper); 

 

/*Gmean (CV%) row*/ 

        ord=2; 

        stat='Geometric LS Mean (CV%)`{super 1}';  

        out=compress(put(round(estimatee,0.01),8.2)); 

        output; 

   /*95% CI row*/ 

        ord=3; 

        stat='95% CI';  

        out=compress(put(floor(100*lowere)/100,8.2))||', 

'||compress(put(ceil(100*uppere)/100,8.2));  

        output; 

    end; 

 if b then do; 

    /*Back transformation*/ 

    estimatee=exp(estimate); 



    lowere=exp(lower)*100;  

    uppere=exp(upper)*100; 

    geocv=sqrt((exp(residual)-1)*100**2);  

/*Gmean (CV%) row*/ 

        ord=2; 

        stat='Geometric LS Mean (CV%)`{super 1}';  

        out=compress(put(round(100*estimatee,0.01),8.2))||' 

('||compress(put(ceil(100*geocv)/100,8.2))||')';    

        output; 

   /*95% CI row*/ 

        ord=3; 

        stat='95% CI';  

        out=compress(put(floor(100*lowere)/100,8.2))||', 

'||compress(put(ceil(100*uppere)/100,8.2));  

        output; 

    end; 

run; 

 

 

 

 

/*Tmax analysis*/ 

 

data tpeak; 

set adppmod(where=(index(paramcd,'TMAX'))); 

run;  

title3 j=l 'Proc Npar1way procedure'; 

 

proc npar1way hl alpha=.05 data=tpeak; 

   by paramn; 

   class trtan; 

   var aval; 

   ods output HodgesLehmann=HodgesLehmann; 

run; 

 

proc univariate data=tpeak noprint; 

   by paramn; 

   class trtan; 

   var aval; 

    output out=meds   n=n1 median=med; 

run; 

 

data tpeakout; 

length out $100 stat $100; 

   set meds(in=a) HodgesLehmann(in=b) ; 

   tpeak=1; 

   /*ordering columns of treatments*/ 

   if b then colord=3; 

   else if trtan=1 then colord=1; 

   else if trtan=2 then colord=2; 

/*columns 1 and 2*/ 

   /* N row*/ 

    if a then do; 

        ord=1; 



        stat='n'; 

        out=compress(put(n1,best.)); 

        output; 

/*Median row*/ 

        ord=2; 

        stat='Median';  

        out=compress(put(round(med,0.01),8.2));    

        output; 

     end; 

/*Column 3*/ 

     if b then do; 

/*Median row*/ 

        ord=2; 

        stat='Median';  

        out=compress(put(round(shift,0.01),8.2));    

        output; 

   /*95% CI row*/ 

        ord=3; 

        stat='95% CI`{super 2}';  

        out=compress(put(floor(100*lowercl)/100,8.2))||', 

'||compress(put(ceil(100*uppercl)/100,8.2));  

        output; 

    end; 

run; 

 

data out; 

   set tabout tpeakout; 

run; 

 

/*transpose for output*/ 

proc sort data=out;  

    by paramn ord colord tpeak; 

run; 

 

proc transpose data=out out=allout(drop=_NAME_) prefix=col; 

   by paramn ord stat tpeak;  

   id colord;  

   var out; 

run; 

 

 

%mixout2(blankn=68, halfblnk=Y,title=Listing 15.4.4.&tabno. &title); 

 

 

/* Standard - macro for paging */ 

%macro outrtf(blankn=68, halfblnk=N, ref=); 

 

/* treatment column headers and footnotes */ 

/*group 1*/ 

 

%let col1=THS 2.2#(N=&tot1);  

%let col2=CC#(N=&tot2); 

%let col3=THS 2.2:CC Ratio#(%); 

 



 

%if &halfblnk=N %then %let halfblnk=; 

%else %if &halfblnk=Y %then %let halfblnk=\~; 

 

/* Standard - just change the number to match the listing you're working 

on. Also change the letters in the*/ 

/* bracket, eg ccb = current cigarette brands. Make sure to do this at 

the top of the code too. */ 

 

 %let tflno=T_15_02_04_&tabno.(PK); 

 

 

       options replace; 

    data table.T_15_02_04_&tabno.; 

    set allout; 

    run; 

    options noreplace; 

 

/* Standard - leave this */ 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

/* Standard - leave this */ 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 

  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

run; 

 

/*page numbers*/ 

data paging; 

    set allout end=a; 

    if tpeak=1 then page=2; 

    else page=1; 

    if a then call symput("tpage",compress(put(page,best.)));  

run; 

 

 

/* Standard - leave this */ 

options number nodate orientation=landscape papersize=&p_pgsize missing=' 

' NOQUOTELENMAX/*turn off warnings about quoted strings too long*/; 

ods escapechar='`'; 

%let linetop = \brdrt\brdrs\brdrw30;  * needs to be 1.5pt so calculated 

in twips (1/20 pt) ; 

%let linebot = \brdrb\brdrs\brdrw30; 

%let linebot2 = \brdrb\brdrs\brdrw15; 

 

 

ods path stdlib.t106324 (read) ; 

ods results off; 

ods rtf toc_data/* contents*/ 

file="/cvn/projects/prj/data/000000106324/TFL/&TFL_Part./&tflno..rtf" 

style=t106324 startpage=yes headery=1440 footery=1440 ; 

ods noproctitle; 

 



%do i=1 %to &tpage; 

 

ODS PROCLABEL = ' '; 

title ; 

footnote; 

%let wd=0; 

 

 

data comp; 

     set paging end=eof; 

     by paramn ord; 

  where page=&i;  

     flag=1; 

       if tpeak=1 then call symput('COL3', 'THS 2.2 - CC');  

     call symput('grp',compress(put(analgr1n,best.)));   

     /* Amend title as needed */ 

     _firtitl="Table 15.2.4.&tabno &title"; 

  _upcas=(length(_firtitl)-

length(compress(_firtitl,'ABCDEFGHIJKLMNOPQRSTUVWXYZ')))/2; 

     len=&blankn.-length("(Page &i of &tpage)"); 

  if eof then do; 

     call symput('_FSRTITL', trim(left(_firtitl))); 

  call symput('_blankn', compress(put(len,best.))); 

  end; 

  drop _firtitl  _upcas len; 

run; 

 

ods listing close; 

 

 

* most set up in template others below; 

* title arial 12pt bold with 12pt paragraph space below; 

* all headers to be arial 11pt bold; 

* data arial 10pt; 

* headers to be central, text values left aligned and numeric centered 

around decimal point; 

/* Update with your variables as needed */ 

proc report data = comp headline headskip missing nowd spanrows split = 

'#' 

%IF &I=1 %THEN %DO; CONTENTS=' ' %END; %ELSE %DO; CONTENTS='' %END;;  

 column flag page paramn  ord stat col1 col2 col3;  

     

 define flag / order noprint; 

  define page       / order order = internal noprint; 

    define paramn      / group style={just=left cellwidth=2.5cm} 

"Parameter"; 

    define ord     / order order=internal noprint; 

    define stat       / display style={just=left cellwidth=3cm} 

"Statistic"; 

    define col1         / display style={just=c cellwidth=3cm} 

style(header)={just=center} "&col1"; 

    define col2         / display style={just=c cellwidth=3cm} 

style(header)={just=center} "&col2"; 



    define col3         / display style={just=c cellwidth=3cm} 

style(header)={just=center} "&col3"; 

  

 break after page / page; 

 

      break before flag / page %IF &I=1 %THEN %DO;  

           CONTENTS="&_FSRTITL" %END; %ELSE %DO; CONTENTS='' %END;;  

 

 compute before page / style={protectspecialchars=off}; 

     line "&linetop"; 

  endcomp; 

 

  compute before _page_ / style={just=left protectspecialchars=off};  

     line "\b\fs24\sa24&_FSRTITL." ;  * \b = bold, \fs24 is font 

size 12pt, \sa24 is space after 12pt; 

   line "&linebot"; 

  endcomp; 

 

  compute after paramn; 

     line " "; 

   endcomp; 

 

compute after _page_ / style={just=left protectspecialchars=off 

pretext="&linetop."}; 

        line "Note:  CC = Conventional cigarettes; THS = Tobacco Heating 

System."; 

LINE 'Note: 1: Geometric LS Mean and 95% CI are the adjusted geometric 

least squares means based on an ANOVA model conducted on log-transformed 

values with study arm, sex and CC consumption reported at screening as 

fixed effect factors. Geometrical CV% of the ratio is estimated from the 

residual mean squares.'; 

LINE "Note: 2: 95% CI are estimated only for the median difference based 

on the Hodges-Lehmann estimate.";  

       line ""; 

        line "Appendix &ref."; 

  line "Path: &TFLpath." &_blankn.*"\~\~" "(Page &i of 

&tpage)"; 

  line "Program Run: &sysdate  &sysuserid   Program Status: 

&status"; 

  endcomp; 

 

run; 

%end; 

ods rtf close; 

ods results on; 

ods path reset; 

 

%mend ; 

 

%outrtf(blankn=68, halfblnk=Y, ref=&reference);  

 

 

%mend; 

 



 

%anal(param=NICOTINE, tabno=39,title=%str(Analysis of Plasma Nicotine 

Concentration Parameters on Day 5 - FAS), reference=15.2.4.37); 

 

%anal(param=COTININE, tabno=40,title=%str(Analysis of Plasma Cotinine 

Concentration Parameters on Day 5 - FAS), reference=15.2.4.38); 

 

proc printto ; run; 

 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 


